语法分析

2020年4月17日 17:05

```
Expression:
     Expression ( "&&" | "<" | "+" | "-" | "*" ) Expression
     Expression "[" Expression "]"
     Expression "." "length"
     Expression "." Identifier "(" [ Expression { "," Expression } ] ")"
     IntegerLiteral
     "true"
     "false"
     Identifier
     "new" "int" "[" Expression "]"
     "new" Identifier "(" ")"
     "!" Expression
     "(" Expression ")"
Statement:
     "{" { Statement } "}"
     "if" "(" Expression ")" Statement "else" Statement
"while" "(" Expression ")" Statement
"System.out.println" "(" Expression ")" ";"
     Identifier "=" Expression ";"
     Identifier "[" Expression "]" "=" Expression ";"
Type:
     "int" "[" "]"
     "boolean"
     "int"
     Identifier
MethodDeclaration:
     "public" Type Identifier "(" [ Type Identifier { "," Type Identifier } ]
     ")" "{" { VarDeclaration } { Statement } "return" Expression ";" "}"
VarDeclaration:
    Type Identifier ";"
ClassDeclaration:
     "class" Identifier [ "extends" Identifier ] "{" { VarDeclaration }
     { MethodDeclaration } "}"
MainClass:
     "class" Identifier "{" "public" "static" "void" "main" "(" "String" "[" "]"
     Identifier ")" "{" Statement "}" "}"
Goal:
    MainClass { ClassDeclaration } EOF
```